# Maintainer: Felix Yan <felixonmars@archlinux.org>

pkgname=pius
pkgver=3.0.0
pkgrel=5
pkgdesc="PGP Individual User Signer"
arch=('any')
url="https://www.phildev.net/pius/"
license=('GPL')
depends=('python')
checkdepends=('python-pytest')
source=("$pkgname-$pkgver.tar.gz::https://github.com/jaymzh/pius/archive/v$pkgver.tar.gz")
sha512sums=('6ead2a456425e073ae32a37f57716ab7221138af75a13104a3b820730869b8ad9bc16b2b1eaeb88b36078967ff0c6233328e30ac8d53d2a54ea9410704372259')

build() {
  cd pius-$pkgver
  python setup.py build
}

check() {
  cd pius-$pkgver
  python -m pytest
}

package() {
  cd pius-$pkgver
  python setup.py install -O1 --prefix=/usr --root="$pkgdir"
}

# vim:set ts=2 sw=2 et:
